// File: Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Address.sol)

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: data_3DABFL.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

//3DABFL data.

import "./Strings.sol";


abstract contract data_3DABFL{
    string[] private sports = [
        "\'0 -0.5 0 -0.5 0 0.5 0 1 0 -0.5 0 1.57 0 -0.5 0 -0.5\'",
        "\'0 -0.5 0 0.5 0 -0.5 0 1 0 0.5 0 1.57 0 -0.5 0 0.5\'"
    ];
    string[] private size = [
        "\'0.2 2.0 0.2\'",
        "\'0.4 0.3 0.5\'",
        "\'0.03 0.6 0.03\'",
        "\'0.03 0.6 0.03\'",
        "\'0.3 0.03 0.03\'",
        "\'0.3 0.03 0.03\'",
        "\'3 1 0.01\'",
        "\'3.2 0.3 0.01\'",
        "\'3.2 0.3 0.01\'",
        "\'1.6 0.2 0.01\'",
        "\'1.8 1.0 0.01\'",
        "\'1.5 0.2 0.01\'",
        "\'0.2 1.2 0.01\'",
        "\'0.2 0.6 0.01\'",
        "\'0.75 0.75 0.02\'",
        "\'0.3 0.3 0.02\'",
        "\'0.3 0.3 0.02\'",
        "\'0.3 0.3 0.02\'",
        "\'0.2 0.8 0.02\'",
        "\'0.5 0.5 0.02\'",
        "\'0.5 0.5 0.02\'"
    ];
    string[] private XYZ = [
        "\'0 0.5 0\'",
        "\'0 1.1 -0.1\'",
        "\'0.1 1.8 0\'",
        "\'-0.1 1.8 0\'",
        "\'-0.25 2.1 0\'",
        "\'0.25 2.1 0\'",
        "\'1.7 0.7 0\'",
        "\'2.2 1.65 0\'",
        "\'1.9 1.35 0\'",
        "\'1 0.1 0\'",
        "\'1.1 -0.5 0\'",
        "\'1.1 -1.1 0\'",
        "\'1.4 -1.8 0\'",
        "\'1.0 -1.5 0\'",
        "\'0.8 0.7 0\'",
        "\'1.8 0.7 0\'",
        "\'1.5 1 0\'",
        "\'2.3 0.5 0\'",
        "\'2.9 0.7 0\'",
        "\'0.7 -0.4 0\'",
        "\'1.5 -0.6 0\'",
        "\'-1.7 0.7 0\'",
        "\'-2.2 1.65 0\'",
        "\'-1.9 1.35 0\'",
        "\'-1 0.1 0\'",
        "\'-1.1 -0.5 0\'",
        "\'-1.1 -1.1 0\'",
        "\'-1.4 -1.8 0\'",
        "\'-1.0 -1.5 0\'",
        "\'-0.8 0.7 0\'",
        "\'-1.8 0.7 0\'",
        "\'-1.5 1 0\'",
        "\'-2.3 0.5 0\'",
        "\'-2.9 0.7 0\'",
        "\'-0.7 -0.4 0\'",
        "\'-1.5 -0.6 0\'"
    ];

    using Strings for uint256;
    string internal _Sugar;

    function R(string memory s) internal pure returns (uint256) {
        return uint256(keccak256(abi.encodePacked(s)));
    }

    function getDNA(uint256 tokenId) public view returns (string memory) {
        return
            string(
                
                    R(string(abi.encodePacked(_Sugar, tokenId.toString()))).toHexString()

            );
    }

    function getAttr1(uint256 tokenId) public view returns (bool ) {
        uint256 seed = R(string(abi.encodePacked("U", getDNA(tokenId))));
        return bool(seed % 100 < 45);
    }

    function getAttr2(uint256 tokenId) public view returns (bool ) {
        uint256 seed = R(string(abi.encodePacked("M", getDNA(tokenId))));
        return bool(seed % 10 < 4);
    }

    function getAttr3(uint256 tokenId) public view returns (bool) {
        uint256 seed = R(string(abi.encodePacked("D", getDNA(tokenId))));
        return bool(seed % 10 < 3);
    }
    function getColor(uint256 tokenId , uint256 _x) public view returns (string memory) {
        uint256 C1 = R(string(abi.encodePacked("R", _x.toString() , getDNA(tokenId)))) % 100;
        uint256 C2 = R(string(abi.encodePacked("G",_x.toString() , getDNA(tokenId)))) % 100;
        uint256 C3 = R(string(abi.encodePacked("B",_x.toString() , getDNA(tokenId)))) % 100;
        return string(abi.encodePacked("\'0.",C1.toString()," ","0.",C2.toString()," ","0.",C3.toString(),"\'"));
    }
    function getHTML(uint256 tokenId,string memory _X3D)
        internal 
        view
        returns (string memory)
    {
        require(tokenId > 0 && tokenId < 7001, "Token ID invalid");
        string[5] memory Ls1;
        string[7] memory Ls2;
        string[6] memory Ls3;
        uint8 y = 6;
        string memory M_1;
        string memory M_2;
        string memory M_3;
        string memory L_1;
        string memory L_2;
        string memory L_3;
        Ls1[0] = '<html><head><meta http-equiv=\'X-UA-Compatible\' content=\'IE=edge\' /><script type=\'text/javascript\' src=';
        Ls1[1] = _X3D;
        Ls1[2] = '> </script></head><body><x3d width=\'100%\' height=\'100%\'><scene><background skyColor=';
        Ls1[3] = getColor(tokenId , 1024);
        Ls1[4] = '></background> ';
        //Ls1[5] = "Copy the following content into the HTML file and open it for viewing";
        Ls3[5] = '</scene></x3d></body></html> ';

        Ls2[0] = '<transform translation=';
        Ls2[2] = ' visible=\'true\'><shape><appearance><material diffuseColor=';
        Ls2[4] = '></material></appearance><Box size=';
        Ls2[6] = '></Box></shape></transform>';

        Ls3[0] = '<transform DEF=';
        Ls3[1] = ' >  ';
        Ls3[2] = ' </transform><timeSensor DEF=\'Clock\' cycleInterval=\'2.0\' loop=\'true\'></timeSensor><OrientationInterpolator DEF=\'ColumnPath\' key=\'0.0 0.20 0.6 1.0\' keyValue=';
        Ls3[3] = '></OrientationInterpolator><Route fromNode=\'Clock\' fromField=\'fraction_changed\' toNode=\'ColumnPath\' toField=\'set_fraction\'></Route><Route fromNode=\'ColumnPath\' fromField=\'value_changed\' toNode=';
        Ls3[4] = ' toField=\'set_rotation\'></Route>';
        for (uint8 i = 0; i < 36; i++) {
            Ls2[1]  = XYZ[i];
            Ls2[3] = getColor(tokenId , i);
            if (i<6){
                Ls2[5] = size[i];
                M_1 = string(abi.encodePacked(M_1,Ls2[0],Ls2[1],Ls2[2]));
                M_1 = string(abi.encodePacked(M_1,Ls2[3],Ls2[4],Ls2[5],Ls2[6]));
            }else if(i<21){
                Ls2[5] = size[i];
                if(i==7){
                    if(getAttr1(tokenId)){
                        M_2 = string(abi.encodePacked(M_2,Ls2[0],Ls2[1],Ls2[2],Ls2[3]));
                        M_2 = string(abi.encodePacked(M_2,Ls2[4],Ls2[5],Ls2[6]));
                    }else{M_2 = string(abi.encodePacked(M_2,' '));}
                }else if(i==12){
                    if(getAttr3(tokenId)){
                        M_2 = string(abi.encodePacked(M_2,Ls2[0],Ls2[1],Ls2[2],Ls2[3]));
                        M_2 = string(abi.encodePacked(M_2,Ls2[4],Ls2[5],Ls2[6]));
                    }else{M_2 = string(abi.encodePacked(M_2,' '));}
                }else if(i==13){
                    if(getAttr2(tokenId)){
                        M_2 = string(abi.encodePacked(M_2,Ls2[0],Ls2[1],Ls2[2],Ls2[3]));
                        M_2 = string(abi.encodePacked(M_2,Ls2[4],Ls2[5],Ls2[6]));
                    }else{M_2 = string(abi.encodePacked(M_2,' '));}
                }else{M_2 = string(abi.encodePacked(M_2,Ls2[0],Ls2[1],Ls2[2],Ls2[3]));
                 M_2 = string(abi.encodePacked(M_2,Ls2[4],Ls2[5],Ls2[6]));}
            }else if(i>20){
                Ls2[5] = size[y];
                if(i==22){
                    if(getAttr1(tokenId)){
                        M_3 = string(abi.encodePacked(M_3,Ls2[0],Ls2[1],Ls2[2],Ls2[3]));
                        M_3 = string(abi.encodePacked(M_3,Ls2[4],Ls2[5],Ls2[6]));
                    }else{M_3 = string(abi.encodePacked(M_3,' '));}
                }else if(i==27){
                    if(getAttr3(tokenId)){
                        M_3 = string(abi.encodePacked(M_3,Ls2[0],Ls2[1],Ls2[2],Ls2[3]));
                        M_3 = string(abi.encodePacked(M_3,Ls2[4],Ls2[5],Ls2[6]));
                    }else{M_3 = string(abi.encodePacked(M_3,' '));}
                }else if(i==28){
                    if(getAttr2(tokenId)){
                        M_3 = string(abi.encodePacked(M_3,Ls2[0],Ls2[1],Ls2[2],Ls2[3]));
                        M_3 = string(abi.encodePacked(M_3,Ls2[4],Ls2[5],Ls2[6]));
                    }else{M_3 = string(abi.encodePacked(M_3,' '));}
                }else{  M_3 = string(abi.encodePacked(M_3,Ls2[0],Ls2[1],Ls2[2],Ls2[3]));
                        M_3 = string(abi.encodePacked(M_3,Ls2[4],Ls2[5],Ls2[6]));}
                y+=1;  
            }
            if (i==20){
                L_1 = string(abi.encodePacked(Ls3[0],"wj",Ls3[1],M_2,Ls3[2],sports[0],Ls3[3],"wj",Ls3[4]));
            }
            if(i==35){
                L_2 = string(abi.encodePacked(Ls3[0],"cQ",Ls3[1],M_3,Ls3[2],sports[1],Ls3[3],"cQ",Ls3[4]));
                L_3 = string(abi.encodePacked(Ls1[0],Ls1[1],Ls1[2],Ls1[3]));
                L_3 = string(abi.encodePacked(L_3,Ls1[4],M_1,L_1,L_2,Ls3[5]));
            }
        }
        return string(abi.encodePacked("Copy the following content into the HTML file and open it for viewing:,",L_3));
    }
}

// File: ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}

// File: IERC721Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Metadata.sol)

pragma solidity ^0.8.0;

import "./IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: NFT_3DABFL.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;
/*
*3D animated butterfly on the chain -NFT.
*All 3D animation data has been uploaded on the chain.
*Get tokenURI Copy the following content into the HTML file and open it for viewing.
*We uploaded x3dom.js to IPFS, and at the same time ensured that the IPFS connection is valid, 
this contract has joined DAO, and the IPFS link of x3dom.js can be revision by voting on the chain.
*We look forward to more Ethereum NFTs that magically interact with other IPFS libraries.
*TG:t.me/NFT_3dabfl
*/
import "./IERC721Receiver.sol";
import "./IERC721Metadata.sol";
import "./Address.sol";
import "./Context.sol";
import "./ERC165.sol";
import "./IERC721.sol";
import "./data_3DABFL.sol";

contract ERC721 is Context, ERC165, IERC721, IERC721Metadata,data_3DABFL{
    using Address for address;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    //X3DOM.JS IPFS URL
    string public X3D;

    address private Admin;

    uint160 public totalSupply; 

    //Counting
    uint256 public X3DDAO;

    uint public Dao;

    

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;


    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;



    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    mapping(address => bool) public mintLock;

    mapping(uint256 => mapping(address => bool)) public DaoS;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;


    constructor(
        string memory name_,
        string memory symbol_,
        string memory sugar_
    ) {
        
        _name = name_; //3D animated butterfly on the chain -NFT
        _symbol = symbol_; //3DABFL
        _Sugar = sugar_;
        Admin = _msgSender();
        X3D = "https://ipfs.io/ipfs/QmZePfriotGLpzkeJgdqij4QwgNHuqxJjRfS16UDHM4Cqc"; //X3Dom
        
    }

    //vote
    function NFTDao() public {
        require(DaoS[X3DDAO][_msgSender()] == false, "No right");
        Dao += _balances[_msgSender()];
        DaoS[X3DDAO][_msgSender()] = true;
    }

    //ReviseX3D ipfs url
    function ReviseX3D(string memory _X3D) public {
        if (totalSupply < 4000) {
            require(_msgSender() == Admin, "Not an admin");
            X3D = _X3D;
        } else if (_msgSender() == Admin && Dao > totalSupply/4) {
            X3D = _X3D;
        } else if (Dao > totalSupply/2 && _balances[_msgSender()] > 5) {
            X3D = _X3D;
            delete Dao;
            X3DDAO += 1;
        }
    }

    function TransferADMIN(address Recipient) public  {
        require(_msgSender() == Admin, "Not an admin");
        Admin = Recipient;
    }


    //claim NFT
    function claim(uint256 tokenId) public {
        require(mintLock[_msgSender()] == false, "Lock");
        require(tokenId > 0 && tokenId < 6301, "Token ID invalid");
        mintLock[_msgSender()] = true;
        _safeMint(_msgSender(), tokenId);
    }
    //admin NFT
    function adminClaim(uint256 tokenId) public {
        require(tokenId > 6300 && tokenId < 7001, "Token ID invalid");
        _safeMint(Admin, tokenId);
    }

    ////Get tokenURI Copy to HTML to open and enjoy.
    function tokenURI(uint256 tokenId) public view virtual override
        returns (string memory)
    {
        require(_exists(tokenId), "ERC721: tokenID is empty");
        return getHTML(tokenId,X3D);
    }




    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId)
        public
        view
        virtual
        override(ERC165, IERC165)
        returns (bool)
    {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner)
        public
        view
        virtual
        override
        returns (uint256)
    {
        require(
            owner != address(0),
            "ERC721: balance query for the zero address"
        );
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId)
        public
        view
        virtual
        override
        returns (address)
    {
        address owner = _owners[tokenId];
        require(
            owner != address(0),
            "ERC721: owner query for nonexistent token"
        );
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }


    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId)
        public
        view
        virtual
        override
        returns (address)
    {
        require(
            _exists(tokenId),
            "ERC721: approved query for nonexistent token"
        );

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved)
        public
        virtual
        override
    {
        _setApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator)
        public
        view
        virtual
        override
        returns (bool)
    {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(
            _isApprovedOrOwner(_msgSender(), tokenId),
            "ERC721: transfer caller is not owner nor approved"
        );

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public virtual override {
        require(
            _isApprovedOrOwner(_msgSender(), tokenId),
            "ERC721: transfer caller is not owner nor approved"
        );
        _safeTransfer(from, to, tokenId, _data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `_data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(
            _checkOnERC721Received(from, to, tokenId, _data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId)
        internal
        view
        virtual
        returns (bool)
    {
        require(
            _exists(tokenId),
            "ERC721: operator query for nonexistent token"
        );
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner ||
            getApproved(tokenId) == spender ||
            isApprovedForAll(owner, spender));
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, _data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        totalSupply += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);

        _afterTokenTransfer(address(0), to, tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(
            ERC721.ownerOf(tokenId) == from,
            "ERC721: transfer from incorrect owner"
        );
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);

        _afterTokenTransfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Approve `operator` to operate on all of `owner` tokens
     *
     * Emits a {ApprovalForAll} event.
     */
    function _setApprovalForAll(
        address owner,
        address operator,
        bool approved
    ) internal virtual {
        require(owner != operator, "ERC721: approve to caller");
        _operatorApprovals[owner][operator] = approved;
        emit ApprovalForAll(owner, operator, approved);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) private returns (bool) {
        if (to.isContract()) {
            try
                IERC721Receiver(to).onERC721Received(
                    _msgSender(),
                    from,
                    tokenId,
                    _data
                )
            returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert(
                        "ERC721: transfer to non ERC721Receiver implementer"
                    );
                } else {
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}
}

// File: Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}